

Personalized Risk Graphs for Intervention Trials – in Technicolor! Using PROC GKPI and CALL EXECUTE to Get the Job Done

Janet Grubber, VA Health Services Research and Development (HSR&D), Durham, NC
Cynthia J. Coffman, VA HSR&D and Duke University Medical Center, Durham, NC
Will Yancy, VA HSR&D and Duke University Medical Center, Durham, NC
Corrine Voils, VA HSR&D and Duke University Medical Center, Durham, NC

ABSTRACT

Research investigators and clinicians are often interested in producing personalized risk graphs for research participants or patients.

We used the GKPI procedure in SAS version 9.2 to produce batches of risk graphs for a randomized trial of an intervention to communicate participants' individual risks for developing type 2 diabetes based on demographic, clinical, and genetic risk factors. We pilot-tested a variety of potential graphs with 35 participants to determine, based on participant feedback, which graphs were most easily understood and most clearly communicated risk.

Graphs created by PROC GKPI can only be produced by hard coding the data values within the procedure rather than by passing the values to the procedure through the *"DATA=Data Set Name"* option. We present a macro and CALL EXECUTE code to demonstrate a method for producing batches of graphs without having to hard code the data values within the GKPI procedure.

This paper presents the final versions of these risk graphs and the associated SAS code for producing them. It is targeted towards users with an intermediate level knowledge of SAS programming.

INTRODUCTION

Several types of graphs can be produced using PROC GKPI, including slider, bullet, dial, speedometer, and traffic light graphs. We pilot-tested 4 graph types (vertical and horizontal sliders, bullets, and dials) in both color and black and white. Participant feedback was used to select and refine the final graph types to be used for the randomized trial. The participants greatly preferred the presentation of these graphs in color rather than in black and white. They also thought the colors used for traffic lights communicated the appropriate level of risk. That is, participants easily associated the color red with higher risk, the color yellow with caution (e.g., close to the danger zone), and the color green with lower risk (e.g., "good to go!"). In addition, most pilot participants preferred the vertical slider graphs, followed by horizontal sliders, dial, and bullet graphs.

In this paper we present the final combination of graphs that we used for communicating type 2 diabetes risk to the research participants in our randomized trial. Based on the pilot-testing, we chose to produce vertical slider graphs in traffic light colors for our randomized trial participants.

The data that contributed to our graphs were obtained through in-person interviews with research participants (e.g., demographics, height and weight, and family history of type 2 diabetes) and blood draws (e.g., fasting plasma glucose and genotypes for 3 genes associated with increased risk of type 2 diabetes). These data were stored in databases and retrieved by running a SAS program that created a SAS data set containing one observation per research participant. Our goal was to create a SAS program that could be invoked by the project coordinator to run a batch of the risk graphs each week for our genetic counselor to use the following week with participants during their risk counseling session. There did not seem to be a straightforward way to accomplish this task because values for individual graphs created by PROC GKPI must be hard coded directly into the procedure rather than being passed to the procedure through the *"DATA=Data Set Name"* option, as is typical of many SAS procedures. In the end, we were able to use CALL EXECUTE to process all observations in the data set to achieve our goal.

In addition to PROC GKPI code for the graphs, we include a macro and the CALL EXECUTE code that we used to process the records for a batch of participants.

PROC GKPI CODE TO CREATE RISK GRAPHS

Figure 1 is the vertical slider graph for lifetime risk of type 2 diabetes based on age, race, gender, and body mass index. It is created by directly entering the desired values (highlighted in yellow) into the PROC GKPI procedure (see Figure 1 Source Code). In addition, we added title statements to identify the graph and ODS language to output the graph to an .rtf file.

12:37 Monday, November 14, 2011

Genetic Risk Study, ID=99999, 111411

Your Current Lifetime Risk of Diabetes is: 59.6% (HIGH Risk)




Figure 1 Source Code:

```
ods rtf file= "H:\presentations\sugi2011\sas
progs\diabetes_risk_famhx_99999_111411.rtf";

title "Genetic Risk Study, ID=99999, 111411";
title2;
title3 "Your Current Lifetime Risk of Diabetes is: 59.6% (HIGH Risk)";
title4;
title5;

proc gkpi mode=raised;
  vslider actual=59.6 bounds=(0 20 40 60 80 100) /
    colors=("green" "yellow" "red" "red" "red")
    AFONT=(f="Comic Sans MS" h=15pt)
    BFONT=(font="Arial" height=.25in) noavalue;
run;
quit;
ods rtf close;
```

Figure 2 is the vertical slider graph for type 2 diabetes risk based on family history of diabetes. It is created by directly entering the desired values (highlighted in yellow) into the PROC GKPI procedure (see Figure 2 Source Code). As with source code for Figure 1, we added title statements to identify the graph and ODS language to output the graph to an .rtf file.

04:03 Monday, November 14, 2011

Genetic Risk Study, ID=99999, 111411

Your Diabetes Risk Based on Your Family History is: LOW



Figure 2. Vertical Slider Graph of Diabetes Risk Based on Family History

Figure 2 Source Code:

```
ods rtf file= "H:\presentations\sugi2011\sas progs
diabetes_risk_famhx_99999_111411.rtf";

title "Genetic Risk Study, ID=99999, 111411";
title2;
title3 "Your Diabetes Risk Based on Your Family History is: LOW";
title4;
title5;

proc gkpi mode=raised;
vslider actual=5 bounds=(0 10 20 30) /
  colors=("green" "yellow" "red")
  AFONT=(f="Comic Sans MS" h=15pt)
  BFONT=(font="Arial" height=.25in) noavalue nobvalue;
run;
quit;
ods rtf close;
```

Figure 3 is the vertical slider graph for type 2 diabetes risk based on fasting plasma glucose level. It is created by directly entering the desired values (highlighted in yellow) into the PROC GKPI procedure (see Figure 3 Source Code). As with source code for Figures 2 and 3, we added title statements to identify the graph and ODS language to output the graph to an .rtf file.

04:03 Monday, November 14, 2011

Genetic Risk Study, ID=99999, 111411

Your Current (color triangle) Blood Sugar is: 99 mg/dL (LOW Risk)




Figure 3. Vertical Slider Graph for Current Fasting Plasma Glucose Level:

Figure 3 Source Code:

```
ods rtf file= "H:\presentations\sugi2011\sas progs\fp9_99999_111411.rtf";

title "Genetic Risk Study, ID=99999, 111411";
title2;
title3 "Your Current (color triangle) Blood Sugar is: 99 mg/dL (LOW Risk)";
title4;
title5;

proc gkpi mode=raised;
  vslider actual=99 bounds=(50 60 70 80 90 100 110 120 125) /
    colors=("green" "green" "green" "green" "green" "yellow" "red" "red")
    AFONT=(f="Comic Sans MS" h=15pt)
    BFONT=(font="Arial" height=.25in) noavalue nolowbound;
run;
quit;
ods rtf close;
```

Figure 4 is the vertical slider graph for the number of copies of high risk alleles for type 2 diabetes. It is created by directly entering the desired values (highlighted in yellow) into the PROC GKPI procedure (see Figure 4 Source Code). As with source code for Figures 1-3, we added title statements to identify the graph and ODS language to output the graph to an .rtf file.

04:08 Monday, November 14, 2011

Genetic Risk Study, ID=99999, 111411

The Number of Copies of High Risk Diabetes Genes that You Have is: 3 (MODERATE Risk)




Figure 4. Vertical Slider Graph for Number of Copies of High Risk Diabetes Genes

Figure 4 Source Code:

```
ods rtf file= "H:\presentations\sugi2011\sas
progs\high_risk_genes_99999_111411.rtf";

title "Genetic Risk Study, ID=99999, 111411";
title3 "The Number of Copies of High Risk Diabetes Genes that You Have is: 3
(MODERATE Risk)";
title4;

proc gkpi mode=raised;
  vslider actual=3 bounds=(0 1 2 3 4 5 6) /
    colors=("green" "green" "green" "yellow" "red" "red")
    AFONT=(f="Comic Sans MS" h=15pt)
    BFONT=(font="Arial" height=.25in) noavalue nolowbound
    LFONT=(font="Arial" height=0.2in);
run;
quit;
ods rtf close;
```

USING MACROS AND CALL EXECUTE TO PROCESS MULTIPLE RECORDS AT ONE TIME AND OUTPUT SEPARATE FILES FOR EACH PARTICIPANT

To run the above set of graphs for multiple participants using an existing data set requires additional steps:

1. Create a macro (e.g. “rungkpi” created in the source code below) that includes all the PROC GKPI procedures for the desired risk graphs. (This macro must identify and call all the variables needed as input variables for all of the PROC GKPI procedures. See source code below.)
2. After creating the macro, set the data set that contains the input variables under a “data _null_” statement. (See source code below.)
3. Use “CALL EXECUTE” to call the macro. This set of commands will output a separate .rtf file with a complete set of graphs for each participant. The output file will include “today’s date” and the ID of the participant in the filename (e.g. diabetes_risk_graphs_99999_111411.rtf, etc.)

Source Code:

```
%let mdyrpt=%SYSFUNC (DATE (), mmdyy6.);

data one;
  input  id crisk clifetimerisktxt $ fmhxtxt $ fmhxgval glucose
         glucoserisktxt $ genes genesrisktxt $;
cards;
99999 59.6 HIGH LOW 5 99 LOW 3 MODERATE
88888 30.1 MODERATE HIGH 25 123 HIGH 5 HIGH
;

run;

*creating macro to run multiple graphs for a given subject;

%macro rungkpi (id,crisk,clifetimerisktxt,fmhxtxt,fmhxgval,glucose,
                glucoserisktxt,genes,genesrisktxt);

ods rtf file= "C:\temp\diabetes_risk_graphs_&id._&mdyrpt..rtf";

***Lifetime Risk of Diabetes Graph;

  title "Genetic Risk Study, ID=&id, &mdyrpt";
  title2;
  title3 "Your Current Lifetime Risk of Diabetes is:  &crisk.% (&clifetimerisktxt.
Risk) ";
  title4;
  title5;

  proc gkpi mode=raised;
    vslider actual=&crisk  bounds=(0 20 40 60 80 100) /
      colors=("green" "yellow" "red" "red" "red")
      AFONT=(f="Comic Sans MS" h=15pt)
      BFONT=(font="Arial" height=.25in) noavalue;
  run;
quit;
```

```

***Family History Risk Graph;

    title "Genetic Risk Study, ID=&id, &mdyrpt";
    title2;
    title3 "Your Diabetes Risk Based on Your Family History is:  &fmhxtxt.";
    title4;
    title5;

    proc gkpi mode=raised;
        vslider actual=&fmhxgval bounds=(0 10 20 30) /
            colors=("green" "yellow" "red")
            AFONT=(f="Comic Sans MS" h=15pt)
            BFONT=(font="Arial" height=.25in) noavalue nobvalue;
    run;
    quit;

***Blood Sugar Risk Graph;

    title "Genetic Risk Study, ID=&id, &mdyrpt";
    title2;
    title3 "Your Current (color triangle) Blood Sugar is:  &glucose mg/dL
(&glucoserisktxt. Risk)";
    title4;
    title5;

    proc gkpi mode=raised;
        vslider actual=&glucose bounds=(50 60 70 80 90 100 110 120 125) /
            colors=("green" "green" "green" "green" "green" "yellow" "red" "red")
            AFONT=(f="Comic Sans MS" h=15pt)
            BFONT=(font="Arial" height=.25in) noavalue nolowbound;
    run;
    quit;

***High Risk Diabetes Genes Graph;

title "Genetic Risk Study, ID=&id, &mdyrpt";
    title2;
    title3 "The Number of Copies of High Risk Diabetes Genes that You Have is:  &genes
(&genesrisktxt. Risk)";
    title4;
    title5;

    proc gkpi mode=raised;
        vslider actual=&genes bounds=(0 1 2 3 4 5 6) /
            colors=("green" "green" "green" "yellow" "red" "red")
            AFONT=(f="Comic Sans MS" h=15pt)
            BFONT=(font="Arial" height=.25in) noavalue nolowbound
            LFONT=(font="Arial" height=0.2in);

    run;
    quit;

ods rtf close;
title;
title3;
run;
quit;

%mend;

```

```
*execute macro %rungskpi for all observations;
data _null_;
  set one;
  call execute('%RUNGKPI('
               ||id ||','
               || crisk ||','
               || clifetimerisktxt || ','
               || fmhxtxt || ','
               || fmhxgval || ','
               || glucose || ','
               || glucoserisktxt || ','
               || genes || ','
               || genesrisktxt||
               ')');
run;
quit;
```

CONCLUSION

SAS PROC GKPI, together with macros and CALL EXECUTE, can be used to create high quality, personalized risk graphs for multiple individuals in a data set. The use of traffic light colors in the graphs easily communicates risk levels to research participants. Although this paper focuses on the use of these methods to produce risk graphs for an intervention trial, the same methods can easily be applied to a wide variety of content areas for an even wider variety of purposes.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Janet Grubber
Enterprise: VA HSR&D
Address: 508 Fulton St. (152)
City, State ZIP: Durham, NC 27705
Work Phone: 919-286-0411 x4080
Fax: 919-416-5836
E-mail: janet.grubber@duke.edu

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.